TORICS IN *MACAULAY2*
TUTORIAL #2

2. Embeddings and Coherent Sheaves

2.1. **Maps to Projective Space.** For which line bundles do the associated complete linear series define a closed embedding?

(a) For `u` equal to `{1,1}`, `{2,1}`, `{1,0}`, `{0,1}`, and `{-1,1}`, execute the following commands in *Macaulay2*.

```
needsPackage "FourTiTwo";
needsPackage "NormalToricVarieties";
Y = hirzebruchSurface 3;
S = ring Y;
M = basis(u,S)
m = rank source M -1;
R = QQ[y_0..y_m];
phi = map(S,R,M);
I = time ker phi;
netList I_*
m - codim I

A = transpose matrix apply((ideal M)_*, e -> first exponents e)
J = time toricMarkov(A,R)
J == I
```

What have you computed? For which values of `u` is the map `phi` a closed embedding? Are there any reasons to prefer the calculation of `I` to that of `J`?

(b) What do the following commands in *Macaulay2* do?

```
X = kleinschmidt(3,{1,2});
S = ring X;
C = intersection for g in (ideal X)_* list (
  posHull transpose matrix for e in gens S list (
    if g % e == 0 then degree e else continue))
rays C
```

By repeating these calculations of other Kleinschmidt varieties, determine which line bundles yield closed embeddings.

2.2. **Canonical Line Bundles.** What is the canonical line bundle on a smooth projective toric variety? For `X` equal to $\mathbb{P}^3$, `hirzebruchSurface 3`, and `smoothFanoToricVariety(4,50)`, execute the following commands in *Macaulay2*.

```
d = dim X
Omega = cotangentSheaf X
prune exteriorPower(d,Omega)
```

Date: 17 June 2014.
2010 *Mathematics Subject Classification.* 14M25, 14Q99.

```
S = ring X;
sheaf(X, S^{ - sum degrees S})
K = OO toricDivisor X
HH^d(X,K)
```

What have you computed? These calculations demonstrate at least two different descriptions for the canonical line bundle; what are they?

2.3. **Cohomology.** How does one compute the cohomology groups for coherent sheaves?

(a) Execute the following commands in *Macaulay2*.

```
PP = projectiveSpace 2;
for i to 10 list HH^0(PP,OO_PP(i))
for i to 10 list HH^2(PP,OO_PP(-i))
matrix table(reverse toList(0..2),toList(-10..10),
  (i,j) -> rank HH^i(PP,OO_PP(j)))
```

What have you calculated? Can you characterize when the cohomology groups vanish? For which line bundles, do all the cohomology groups vanish? How are the groups $H^0\big(\mathbb{P}^2, \mathscr{O}_{\mathbb{P}^2}(i)\big)$ related to the Cox ring? How are the groups $H^2\big(\mathbb{P}^2, \mathscr{O}_{\mathbb{P}^2}(i)\big)$ related to the module $\mathbb{C}[x_0^{-1}, x_1^{-1}, x_2^{-1}] \cdot x_0^{-1} x_1^{-1} x_2^{-1}$?

(b) Execute the following commands in *Macaulay2*.

```
Y = hirzebruchSurface 2;
matrix table(reverse toList(0..5),toList(-10..10),
  (i,j) -> rank HH^0(Y, OO_Y(j,i)))
matrix table(reverse toList(-5..5),toList(-10..10),
  (i,j) -> rank HH^1(Y, OO_Y(j,i)))
matrix table(reverse toList(-5..5),toList(-10..10),
  (i,j) -> rank HH^2(Y, OO_Y(j,i)))
```

What have you calculated? Can you characterize when the cohomology groups vanish? For which line bundles, do all the cohomology groups vanish? How are the groups $H^0\big(Y, \mathscr{O}_Y(i,j)\big)$ related to the Cox ring? How are the groups $H^1\big(Y, \mathscr{O}_Y(i,j)\big)$ related to the module $\mathbb{C}[x_0^{-1}, x_1, x_2^{-1}, x_3] \cdot x_0^{-1} x_2^{-1} \oplus \mathbb{C}[x_0, x_1^{-1}, x_2, x_3^{-1}] \cdot x_1^{-1} x_2^{-1}$? How are the groups $H^2\big(Y, \mathscr{O}_Y(i,j)\big)$ related to the module $\mathbb{C}[x_0^{-1}, x_1^{-1}, x_2^{-1}, x_3^{-1}] \cdot x_0^{-1} x_1^{-1} x_2^{-1} x_3^{-1}$? How does Serre duality manifest in these calculations?

References

[M2] D.R. Grayson and M.E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at www.math.uiuc.edu/Macaulay2/.

[CLS] D.A. Cox, J.B. Little, and H.K. Schenck, *Toric varieties*, Graduate Studies in Mathematics, vol. 124, American Mathematical Society, Providence, RI, 2011.

This work is licensed under a Creative Commons Attribution-NonCommercial 4.0 International License.

GREGORY G. SMITH, DEPARTMENT OF MATHEMATICS AND STATISTICS, QUEEN'S UNIVERSITY, KINGSTON, ONTARIO, K7L 3N6, CANADA

E-mail address: ggsmith@mast.queensu.ca